

Introduction à l'utilisation de Xcas.

Vous devez tout d'abord télécharger Xcas et choisir le langage Xcas.

Ce TP a pour but de se familiariser avec les commandes usuelles de ce logiciel de calcul formel.

La prise en main dans le menu principal de Xcas est beaucoup plus complète.

Il se peut que Xcas garde en mémoire des valeurs de lignes en aval, si vous faites des aller retours dans les lignes.

Si problème, n'hésitez pas à ouvrir un nouveau fichier, voire à fermer et rouvrir Xcas.

Cliquer sur session puis nouveau pour ouvrir un nouveau fichier Xcas.

L'aide.

Taper ?solve

Le ? qui précède une commande vous envoie sur cette commande dans l'aide.

Vous pouvez aussi utiliser Aide puis index.

La syntaxe.

Taper :

x=2
x
x :=2
x
x+3
5*x+1
X+3

Que remarque-t-on ?

Taper :

x :=2
x
x := 'x'
x+1
x :=3
x
purge(x)
x+5

Que remarque-t-on ?

Taper :

x :=2 ;
x

Que remarque-t-on ?

Calcul algébrique Taper les lignes suivantes :

2+3
2-3
2*3
2^3
2**3
2/3
1/2+1/6
sqrt(4)
root[2](4)
4^(1/2)
sqrt(8)
root[3](8)
ans()^3

Que faut-il retenir?

Valeur approchée et précision: Taper les lignes suivantes :

```
evalf(sqrt(8))
evalf(root[3](8))
evalf(sqrt(2))
digits:=3
evalf(sqrt(2))
digits:=8
evalf(sqrt(2))
evalf(pi)
round(3.45)
round(3.55)
floor(3.55)
floor(3.45)
```

Que faut-il retenir?

Manipulation des variables: Taper les lignes suivantes :

```
z:=x+y
x:=2
y:=4
z
X*y
purge(x,y,z)
x+1
x :=3
x+1
'x'+1
purge(x)
```

Que faut-il retenir?

Développer et simplifier Taper les lignes suivantes :

```
y :=(3*x-2)*(x-3)-2*x^2+4
expand(y)
z := (x+3)*(2*x-5)-(x+3)*(x^2+5)
factor(z)
t:=x^2-9
factor(t)
```

Dans la barre de menu, aller à scolaire, seconde, développer

A la suite de developper(qui vient d'apparaître sur la ligne, taper $(x+2)*(x-5)$)

Dans la barre de menu, aller à scolaire, seconde, factoriser

A la suite de factoriser(qui vient d'apparaître sur la ligne, taper x^2-9)

Dans la barre de menu, aller à scolaire, seconde, simplifier

A la suite de simplifier(qui vient d'apparaître sur la ligne, taper $(x+5)*(2*x-3)+(x+6)*(1-x)$)

Cool , isn't it ?

Résolution d'équations Taper les lignes suivantes :

```
solve(2*x-4,x)
solve(2*x-3,x)
fsolve(2*x-3,x)
solve(x^2-5,x)
fsolve(x^2-5,x)
solve(x^2-5*x+3,x)
fsolve(x^2-5*x+3,x)
```

```
proot(x^2-5*x+3)
proot(x^3-5*x^2+3*x+1)
```

Plus hard:

```
solve(cos(x)=sqrt(2)/2,x)
assume(x>0)
solve(cos(x)=sqrt(2)/2,x)
assume(x>pi/2)
solve(sin(x)=sqrt(2)/2,x)
```

Systèmes: Taper les lignes suivantes, après avoir ouvert un nouveau fichier :

```
solve([2*x-3*y=-4,x+2*y=4],[x,y])
solve([2*x-3*y=5,x+2*y=4],[x,y])
solve([2*x-3*y=5, 2*x-3*y=6],[x,y])
solve([2*x-3*y=5, 2*x-3*y=5],[x,y])
```

Que faut-il retenir de ce TP ?